

# Anova 3-way Interactions: Deconstructed

2008 Winter Stata Users Group

Phil Ender

UCLA Statistical Consulting Group

November 2008

# Three approaches

Explaining 2-way interactions is pretty routine but 3-way interactions can be intimidating to some people. This presentation will look at three approaches to understanding a 3-way interaction:

- ▶ 1 Conceptual Approach
- ▶ 2 Anova Approach
- ▶ 3 Regression Approach

## Meet the data

```
. use http://www.ats.ucla.edu/stat/stata/faq/threeway, clear
```

This is a synthetic dataset for a 2x2x3 factorial anova design with 2 observations per cell. The data were constructed to have different two-way interactions for each level of A.


# Anova table

```
. anova y a b c a*b a*c b*c a*b*c
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
Model	497.8333333	11	45.2575758	33.94	0.0000
a	150	1	150	112.50	0.0000
b	.6666666667	1	.6666666667	0.50	0.4930
c	127.5833333	2	63.7916667	47.84	0.0000
a*b	160.166667	1	160.166667	120.13	0.0000
a*c	18.25	2	9.125	6.84	0.0104
b*c	22.58333333	2	11.2916667	8.47	0.0051
a*b*c	18.58333333	2	9.2916667	6.97	0.0098
Residual	16	12	1.33333333		
<hr/>					
Total	513.8333333	23	22.3405797		


$b^*c$  means plot at  $a1$  – possible interaction

$b^*c$  at  $a=1$



$b^*c$  means plot at  $a=2$  – unlikely interaction

$b^*c$  at  $a=2$



Part 1: Introduction

**Part 2: Conceptual Approach**

Part 3: Anova Approach

Part 4: Regression Approach

Part 5: Determining critical values

# Conceptual Approach

## About the conceptual approach

Basically, this approach involves running separate anovas on subsets of the original model and manually computing the correct F-ratio using the MS residual from the original 3-factor model.

You will need to save the MS residual value from the original anova model.

$$\text{MSresidual} = 1.333333333$$

b\*c at a1

Run 2-way anova at a1

```
. anova y b c b*c if a==1
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
b	70.0833333	1	70.0833333	56.07	0.0003
c	24.6666667	2	12.3333333	9.87	0.0127
b*c	40.6666667	2	20.3333333	16.27	0.0038
Residual	7.5	6	1.25		
<hr/>					
Total	142.916667	11	12.9924242		

F-ratio for b\*c interaction does not use correct error term.

## b\*c at a1 (cont)

Manually compute correct F-ratio for b\*c interaction.

$$F(a*b \text{ at } a1) = 20.33333333 / 1.33333333$$

$$= 15.25$$

b\*c at a2

Repeat 2-way anova at a2

```
. anova y b c b*c if a==2
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
b	90.75	1	90.75	64.06	0.0002
c	121.166667	2	60.5833333	42.76	0.0003
b*c	.5	2	.25	0.18	0.8424
Residual	8.5	6	1.41666667		
<hr/>					
Total	220.916667	11	20.0833333		

Again, F-ratio for b\*c interaction does not use correct error term.

## b\*c at a2 (cont)

Manually compute correct F-ratio for b\*c interaction.

$$F(a*b \text{ at } a1) = 1.41666667 / 1.33333333$$

$$= 0.1875$$

## Summary for b\*c anovas

F-ratio for b\*c at a1 = 15.25

F-ratio for b\*c at a2 = 0.1875


It is likely that the F-ratio for b\*c at a1 is statistically significant while the F-ratio at a2 is not. We will postpone the discussion of critical values until the last section.

## Follow up tests of simple main effects

Since it is likely that the  $b*c$  interaction at  $a1$  will be significant, we will need to follow up with some tests of simple main effects. In this case, we will focus on differences in the levels of  $c$  at  $b1$  and  $b2$  at  $a1$ .

$b^*c$  means plot at  $a=1$

$b^*c$  at  $a=1$



## Test of simple main effects

Oneway anova for c at b1 & a1

```
. anova y c if b==1 & a==1
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
c	64	2	32	16.00	0.0251
Residual	6	3	2		
<hr/>					
Total	70	5	14		

Recompute F-ratio for c using error term from original model.

$$F(c \text{ at } b1 \& a1) = 32 / 1.333333333 = 24$$

## Test of simple main effects (cont)

Oneway anova for c at b2 & a1

```
. anova y c if b==2 & a==1
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
c	1.333333333	2	.6666666667	1.33	0.3852
Residual	1.5	3	.5		
<hr/>					
Total	2.833333333	5	.5666666667		

Recompute F-ratio for c using error term from original model.

$$F(c \text{ at } b1 \& a1) = .6666666667 / 1.333333333 = 0.5$$

## Summary for tests of simple main effects

F-ratio for c at b1, a1 = 24.0

F-ratio for c at b2, a2 = 0.5

It is likely that the F-ratio for differences in c at b1 is statistically significant while F-ratio at b2 is not. We are still postponing the discussion of critical values.

# Anova Approach

## About the anova approach

The anova approach involves running several anova models, creating contrast matrices and using the test command to test the effects of interest.

We could, of course, do this with the original 3-factor model but there are way too many terms to keep track of so, instead, we will do this in several steps using simpler models.

## b\*c at levels of a

```
. anova y b c b*c|a /* b*c is nested in a */
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
Model	497.833333	11	45.2575758	33.94	0.0000
b	.666666667	1	.666666667	0.50	0.4930
c	127.583333	2	63.7916667	47.84	0.0000
b*c a	369.583333	8	46.1979167	34.65	0.0000
Residual		16	1.33333333		
<hr/>					
Total	513.833333	23	22.3405797		

# showorder

```
. test, showorder
```

Order of columns in the design matrix

```
1: _cons
2: (b==1)
3: (b==2)
4: (c==1)
5: (c==2)
6: (c==3)
7: (b==1)*(c==1)*(a==1)
8: (b==1)*(c==1)*(a==2)
9: (b==1)*(c==2)*(a==1)
10: (b==1)*(c==2)*(a==2)
11: (b==1)*(c==3)*(a==1)
12: (b==1)*(c==3)*(a==2)
13: (b==2)*(c==1)*(a==1)
14: (b==2)*(c==1)*(a==2)
15: (b==2)*(c==2)*(a==1)
16: (b==2)*(c==2)*(a==2)
17: (b==2)*(c==3)*(a==1)
18: (b==2)*(c==3)*(a==2)
```

## create contrast matrices for b\*c at levels of a

```
. matrix bc1=(0,0,0,0,0,0,1,0,0,0,-1,0,-1,0,0,0,1,0 ///
               0,0,0,0,0,0,0,0,1,0,-1,0,0,0,-1,0,1,0)

. matrix bc2=(0,0,0,0,0,0,0,1,0,0,0,-1,0,-1,0,0,0,1 ///
               0,0,0,0,0,0,0,0,1,0,-1,0,0,0,-1,0,1)
```

## test b\*c at a1 & b\*c at a2

```
/* test b*c at a==1 */
. test, test(bc1)

( 1) b[1]*c[1]*a[1] - b[1]*c[3]*a[1] - b[2]*c[1]*a[1] + b[2]*c[3]*a[1] = 0
( 2) b[1]*c[2]*a[1] - b[1]*c[3]*a[1] - b[2]*c[2]*a[1] + b[2]*c[3]*a[1] = 0

F( 2,      12) =     15.25
                  Prob > F =    0.0005

/* test b*c at a==2 */
. test, test(bc2)

( 1) b[1]*c[1]*a[2] - b[1]*c[3]*a[2] - b[2]*c[1]*a[2] + b[2]*c[3]*a[2] = 0
( 2) b[1]*c[2]*a[2] - b[1]*c[3]*a[2] - b[2]*c[2]*a[2] + b[2]*c[3]*a[2] = 0

F( 2,      12) =     0.1875
                  Prob > F =    0.8314
```

## c at levels of b – for tests of simple main effects

```
. anova y c|a*b /* c is nested in a*b */
```

Source	Partial SS	df	MS	F	Prob > F
<hr/>					
Model	497.833333	11	45.2575758	33.94	0.0000
c a*b	497.833333	11	45.2575758	33.94	0.0000
Residual		16	1.33333333		
<hr/>					
Total	513.833333	23	22.3405797		

## showorder for c nested in a\*b

```
. test, showorder

Order of columns in the design matrix
1: _cons
2: (c==1)*(a==1)*(b==1)
3: (c==1)*(a==1)*(b==2)
4: (c==1)*(a==2)*(b==1)
5: (c==1)*(a==2)*(b==2)
6: (c==2)*(a==1)*(b==1)
7: (c==2)*(a==1)*(b==2)
8: (c==2)*(a==2)*(b==1)
9: (c==2)*(a==2)*(b==2)
10: (c==3)*(a==1)*(b==1)
11: (c==3)*(a==1)*(b==2)
12: (c==3)*(a==2)*(b==1)
13: (c==3)*(a==2)*(b==2)
```

## create contrast matrices for c nested in a\*b

```
. matrix c1=(0,1,0,0,0,0,0,0,-1,0,0,0\ ///
               0,0,0,0,0,1,0,0,0,-1,0,0,0)

. matrix c2=(0,0,1,0,0,0,0,0,0,-1,0,0\ ///
               0,0,0,0,0,0,1,0,0,0,-1,0,0)
```

test c at b1 & c at b2

```
/* test c at b==1 */
. test, test(c1)

( 1)  c[1]*a[1]*b[1] - c[3]*a[1]*b[1] = 0
( 2)  c[2]*a[1]*b[1] - c[3]*a[1]*b[1] = 0
```

F( 2, 12) = 24.00
Prob > F = 0.0001

```
/* test c at b==2 */
. test, test(c2)

( 1)  c[1]*a[1]*b[2] - c[3]*a[1]*b[2] = 0
( 2)  c[2]*a[1]*b[2] - c[3]*a[1]*b[2] = 0
```

F( 2, 12) = 0.50
Prob > F = 0.6186

# Regression Approach

## About the anova approach

The regression approach involves creating dummy variables for all the main effects and interactions and then testing them in the proper combinations to get the tests of simple interactions and simple main effects.

## Create dummies and interactions

- . recode a (1=0)(2=1)
- . recode b (1=0)(2=1)
- . generate c1=c==1
- . generate c2=c==2
- . generate ab=a\*b
- . generate ac1=a\*c1
- . generate ac2=a\*c2
- . generate bc1=b\*c1
- . generate bc2=b\*c2
- . generate abc1=a\*bc1
- . generate abc2=a\*bc2

# Regression model

```
. regress y a b c1 c2 ab ac1 ac2 bc1 bc2 abc1 abc2, noheader
```

y	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
a	- .5	1.154701	-0.43	0.673	-3.015876 2.015876
b	-9.5	1.154701	-8.23	0.000	-12.01588 -6.984124
c1	-8	1.154701	-6.93	0.000	-10.51588 -5.484124
c2	-4	1.154701	-3.46	0.005	-6.515876 -1.484124
ab	15	1.632993	9.19	0.000	11.44201 18.55799
ac1	0	1.632993	0.00	1.000	-3.557986 3.557986
ac2	1	1.632993	0.61	0.552	-2.557986 4.557986
bc1	9	1.632993	5.51	0.000	5.442014 12.55799
bc2	5	1.632993	3.06	0.010	1.442014 8.557986
abc1	-8.5	2.309401	-3.68	0.003	-13.53175 -3.468247
abc2	-5.5	2.309401	-2.38	0.035	-10.53175 -.4682473
_cons	19	.8164966	23.27	0.000	17.22101 20.77899

## Test of 3-way interaction

```
. test abc1 abc2
```

```
( 1) abc1 = 0
```

```
( 2) abc2 = 0
```

```
F( 2,     12) =      6.97
```

```
Prob > F =    0.0098
```

## Test of b\*c at a1 & b\*c at a2

```
/* test b*c at a1 */
```

```
. test bc1 bc2
( 1) bc1 = 0
( 2) bc2 = 0
```

```
F( 2,     12) =    15.25
Prob > F =    0.0005
```

```
/* test b*c at a2 */
```

```
. test bc1+abc1=0
. test bc2+abc2=0, accum
( 1) bc1 + abc1 = 0
( 2) bc2 + abc2 = 0
```

```
F( 2,     12) =    0.1875
Prob > F =    0.8314
```

## Test of c at b1 & c at b2

```
/* test for c at b==1 & a==1 */
. test c1 c2
( 1) c1 = 0
( 2) c2 = 0
```

F( 2, 12) = 24.00  
Prob > F = 0.0001

```
/* test for c at b==2 & a==1 */
. test c1+bc1=0
. test c2+bc2=0, accum
( 1) c1 + bc1 = 0
( 2) c2 + bc2 = 0
```

F( 2, 12) = 0.50  
Prob > F = 0.6186

# Determining critical values

## Computing critical values

There are at least four methods for computing critical values found in the literature.

- ▶ Dunn's procedure
- ▶ Marascuilo & Levin
- ▶ Per family error rate
- ▶ Simultaneous test procedure (closely related to the Scheffé's multiple comparison procedure)

No clear consensus as to which approach is best.

## critical values for a\*b at two levels of a

```
. smecriticalvalue, number(2) df1(2) df2(12) dfmodel(11)

number of tests: 2
    numerator df: 2
    denominator df: 12
original model df: 11
```

Critical value of F for alpha = .05 using ...

---

Dunn's procedure	= 6.2753765
Marascuilo & Levin	= 7.1335873
per family error rate	= 5.0958672
simultaneous test procedure	= 10.245969

## critical values for a\*b at two levels of a (cont)

Critical value of F for alpha = .05 using ...

---

Dunn's procedure	= 6.2753765
Marascuilo & Levin	= 7.1335873
per family error rate	= 5.0958672
simultaneous test procedure	= 10.245969

Using the critical values from the previous slide, both the F-ratios for b\*c at a1 (15.25) and c at b1, a1 (24.0) were statistically significant regardless of which method was used to determine the critical value.

## download ado-file

Use -findit- command

```
. findit smecriticalvalue
```

Follow installation instructions.

## Reference

(1995) Kirk, R.E. Experimental design: Procedures for the behavioral sciences (3rd ed). Pacific Grove, CA: Brooks/Cole

- Part 1: Introduction
- Part 2: Conceptual Approach
- Part 3: Anova Approach
- Part 4: Regression Approach
- Part 5: Determining critical values

# The End